Intrinisically Typed Syntax That Works (Fresh Perspective)

ANONYMOUS AUTHOR(S)

In dependently typed proof assistants, intrinsically typed syntax promises elegant mechanizations: well-typed terms are represented by well-typed data, substitution preserves typing by construction, and type safety proofs collapse to straightforward induction. For simple calculi such as the simply-typed lambda calculus this promise is fulfilled. For realistic polymorphic systems, however, the technique can become impractical: renamings and substitutions indexed by other substitutions force proofs to transport terms across equal types that are not definitionally equal. This phenomenon—informally known as transfer hell—causes even routine metatheoretic arguments to be dominated by administrative reasoning.

This fresh perspective shows how to make intrinsically typed syntax work in practice for polymorphic calculi. We show that giving substitutions canonical computational behavior (via Agda rewriting) yields a mechanization of System F's metatheory in Agda that avoids explicit transport reasoning.

Our approach simplifies proofs of metatheoretic properties: typing preservation becomes definitional, substitution lemmas and corresponding transfers disappear. Generally, when intrinsic encodings generate transports, the right move is often to strengthen definitional equality rather than add more lemmas. We extract practical guidelines for applying this methodology.

ACM Reference Format:
Anonymous Author(s). 2018. Intrinisically Typed Syntax That Works (Fresh Perspective). *ACM Trans. Program. Lang. Syst.* 37, 4, Article 111 (August 2018), 17 pages. https://doi.org/XXXXXXX.XXXXXXX

1 Introduction

Intrinsically typed syntax is widely recommended for mechanizing type systems in dependently typed proof assistants. By representing well-typed terms as well-typed data, typing invariants hold by construction as in a Church-style presentation, so typing preservation is immediate. For small calculi this approach is highly effective. However, for realistic polymorphic calculi such as System F, mechanizations often become complicated: conceptually simple proofs become cluttered with transports due to administrative equalities.

The simply-typed lambda calculus is the canonical example where intrinsically typed syntax is particularly effective. Figure 1 shows typical Agda definitions of the syntax of types, type contexts, variables, and expressions, which are indexed by contexts and types (left column). Building on these definitions we can define a small-step operational semantics and prove type safety. This is because subject reduction holds by construction and progress follows by a simple inductive proof (right column). We elide the definition of the substitution operator _[_], as we discuss substitutions in detail in Section 2.1.

The intrinsic encoding makes type safety proofs so concise that they can serve as textbook examples [5]. Nothing comparable to transport reasoning arises in this development as only definitional properties of substitutions are required. However, the same approach breaks down for polymorphic calculi and the remainder of this paper explains how to recover some of its simplicity.

Transfer hell is the colloquial term for being forced to apply type-changing lemmas to terms that are already known to be equal. This phenomenon appears, for example, in intrinsic mechanizations of System F: expression substitutions depend on type substitutions and type substitutions depend on renamings; composing them yields types that are equal only propositionally, forcing transports throughout proofs. The root problem is that substitutions that are propositionally equal are not definitionally equal, so terms must repeatedly be transported across equal types.

2018. ACM 1558-4593/2018/8-ART111
https://doi.org/XXXXXXX.XXXXXXX

```
data Type : Set where
  𝟙    : Type
  _⇒_ : Type → Type → Type

data Ctx : Set where
  ∅    : Ctx
  _▷_ : Ctx → Type → Ctx

data _∋_ : Ctx → Type → Set where
  here  : (Γ ▷ T) ∋ T
  there : Γ ∋ T → (Γ ▷ U) ∋ T

data _⊢_ Γ : Type → Set where
  con : Γ ⊢ 𝟙
  var : Γ ∋ T → Γ ⊢ T
  lam : (Γ ▷ T) ⊢ U → Γ ⊢ (T ⇒ U)
  app : Γ ⊢ (T ⇒ U) → Γ ⊢ T → Γ ⊢ U

data _⟶_ {Γ} {T} : Γ ⊢ T → Γ ⊢ T → Set where
  β-lam : app (lam e₁) e₂ ⟶ (e₁ [ e₂ ])
  ξ-app : e₁ ⟶ e₁′ → app e₁ e₂ ⟶ app e₁′ e₂

data Value {Γ} : Γ ⊢ T → Set where
  con : Value con
  lam : (e : (Γ ▷ T) ⊢ U) → Value (lam e)

data Progress (e : Γ ⊢ T) : Set where
  done : Value e → Progress e
  step : e ⟶ e′ → Progress e

progress : (e : ∅ ⊢ T) → Progress e
progress con      = done con
progress (lam e)  = done (lam e)
progress (app e₁ e₂)
  with progress e₁
... | step x      = step (ξ-app x)
... | done (lam e) = step β-lam
```

Fig. 1. Call-by-name simply-typed lambda calculus (adapted from the PLFA textbook [5])

One workaround is to state definitions and proofs with explicit equivalence relations, but this practice is cumbersome, inefficient, and far away from the concision of definitional equality. Our proposal thus shifts the focus to computation on substitutions: if extensionally equal substitutions do not normalize identically, transport reasoning is unavoidable.

Hence, the simple message of this fresh perspective is: when intrinsically typed syntax becomes complicated, the problem is often not the encoding of terms but the definition of equality on substitutions. Strengthening definitional equality can simplify metatheoretical proofs more effectively than adding automation or additional lemmas.

We substantiate our claims with a case study on renaming and substitution in an intrinsic encoding of System F. We strengthen definitional equality by internalizing the equational theory of substitutions (the σ-calculus) using Agda's rewriting mechanism. As a consequence, extensionally equal substitutions normalize to the same form and transports disappear from routine proofs.

Contributions

- We show how to eliminate "transfer hell" in intrinsically typed encodings of System F by internalizing the equational laws of the σ-calculus for substitutions using Agda's rewriting mechanism.
- We demonstrate that the resulting rewrite system is compatible with Agda's definitional equality (in particular, terminating and confluent) for a functional representation of substitutions.
- We mechanize parts of the metatheory of System F, obtaining substantially simpler proofs and an intrinsic treatment of type conversion for the latter.
- From this experience we extract practical guidelines for designing intrinsically typed syntax in dependently typed proof assistants.

```
data Type (n : Nat) : Set where
  `_    : Var n → Type n
  ∀α    : Type (1 + n) → Type n
  _⇒_   : Type n → Type n → Type n
```

```
_ : Type 0 - a closed type:  ∀α. α→α
_ = ∀α (` zero ⇒ ` zero)
_ : Type 0 - ∀αβ. α→β→α
_ = ∀α (∀α (` suc zero ⇒ ` zero ⇒ ` suc zero))
```

Fig. 2. Syntax of System F types (Agda definition left, examples right)

All code fragments in this document are extracted from typechecked Agda code, which is available as a supplement.

Related Work and Positioning

To our knowledge, previous intrinsically typed mechanizations of System F [1, 2, 10] follow a natural design: expression syntax is indexed by typing derivations and substitutions are indexed by substitutions at the type level. While elegant at the level of definitions, this design makes proofs brittle, because they are prone to transfer hell. Our work eliminates transfers resulting from substitutions, thus avoiding transfer hell.

Recent work studies rewrite rules in dependent type theories, including Agda [3] and Rocq [6]. Local rewrite rules are an active research direction [7]. In Rewriting Type Theory (RTT), the reduction relation is extended by turning propositional equalities into reduction rules: a term matched by the left-hand side of an equation reduces to the right-hand side. Rewrites are safe if they preserve subject reduction and consistency. The former is guaranteed by confluence of the combined rewrite system, the latter by only rewriting proved propositional equalities. Our work is based on Agda's implementation of RTT.

Our approach is inspired by the work on Autosubst [11–13]. While they rely on tactics to normalize substitutions, we do so by rewriting [4]. We build on a standard functional representation of renamings and substitutions [1]. The main challenge in setting up the σ-calculus is to fine-tune the reduction behavior of the definitions with the rewrite rules to ensure that the combined reduction relation remains terminating and confluent.

Overview

Section 2 presents our scoped encoding of System F types. It discusses the standard encoding of functional renamings and substitutions (Section 2.2) and introduces the σ-calculus laws (Section 2.3). Section 3 presents the intrinsically typed encoding of System F expressions. It defines expression substitutions (Section 3.3) and a small-step operational semantics (Section 3.4). Section 4 demonstrates a concrete instance of transfer hell, once treated manually and once by using our method. Section 5 outlines our recommendations and guidelines.

2 System F Types

2.1 Syntax

The syntax of System F types uses the standard de Bruijn encoding of variables by numbers. The type Type n (Figure 2) stands for the set of types with n type variables in scope. Henceforth, we call such n a *scope*. Type variables for scope n are drawn from the type Var n, i.e., the set $\{0, \ldots, n-1\}$, written as zero, suc zero, suc (suc zero), ...and ranged over by α. The $\forall\alpha$-binder introduces a new type variable, and $\Rightarrow$ is the infix function type constructor. We let T range over types.

```
-- renamings
Ren : Nat → Nat → Set
Ren n₁ n₂ = Var n₁ → Var n₂

opaque
  -- weakening
  wkᴿ : Ren n (1 + n)
  wkᴿ = suc

  -- identity renaming
  idᴿ : Ren n n
  idᴿ α = α

  -- extend with new variable
  _·ᴿ_ : Var n₂ → Ren n₁ n₂ → Ren (1 + n₁) n₂
  (α ·ᴿ ζ) zero    = α
  (_ ·ᴿ ζ) (suc α) = ζ α

  -- apply renaming to variable
  _&ᴿ_ : Var n₁ → Ren n₁ n₂ → Var n₂
  α &ᴿ ζ = ζ α

  -- left-to-right composition
  _⨟ᴿ_ : Ren n₁ n₂ → Ren n₂ n₃ → Ren n₁ n₃
  (ζ₁ ⨟ᴿ ζ₂) α = ζ₂ (ζ₁ α)

-- lifting
_↑ᴿ : Ren n₁ n₂ → Ren (1 + n₁) (1 + n₂)
ζ ↑ᴿ = zero ·ᴿ (ζ ⨟ᴿ wkᴿ)

-- apply renaming to type
_[_]ᴿ : Type n₁ → Ren n₁ n₂ → Type n₂
(` α) [ ζ ]ᴿ       = ` (α &ᴿ ζ)
(∀α T) [ ζ ]ᴿ      = ∀α (T [ ζ ↑ᴿ ]ᴿ)
(T₁ ⇒ T₂) [ ζ ]ᴿ = (T₁ [ ζ ]ᴿ) ⇒ (T₂ [ ζ ]ᴿ)
```

```
-- substitutions
Sub : Nat → Nat → Set
Sub n₁ n₂ = Var n₁ → Type n₂

opaque
  -- lift renaming to substitution
  ⟨_⟩ : Ren n₁ n₂ → Sub n₁ n₂
  ⟨ ζ ⟩ α = ` (α &ᴿ ζ)

  -- extend with new type
  _·ˢ_ : Type n₂ → Sub n₁ n₂ → Sub (1 + n₁) n₂
  (T ·ˢ η) zero    = T
  (T ·ˢ η) (suc α) = η α

  -- apply substitution to variable
  _&ˢ_ : Var n₁ → Sub n₁ n₂ → Type n₂
  α &ˢ η = η α

  -- lifting
  _↑ˢ : Sub n₁ n₂ → Sub (1 + n₁) (1 + n₂)
  η ↑ˢ = (` zero) ·ˢ λ α → (η α) [ wkᴿ ]ᴿ

-- apply substitution to type
_[_]ˢ : Type n₁ → Sub n₁ n₂ → Type n₂
(` α) [ η ]ˢ       = α &ˢ η
(∀α T) [ η ]ˢ      = ∀α (T [ η ↑ˢ ]ˢ)
(T₁ ⇒ T₂) [ η ]ˢ = (T₁ [ η ]ˢ) ⇒ (T₂ [ η ]ˢ)

opaque
  -- left-to-right composition
  _⨟ˢ_ : Sub n₁ n₂ → Sub n₂ n₃ → Sub n₁ n₃
  (η₁ ⨟ˢ η₂) α = (η₁ α) [ η₂ ]ˢ
```

Fig. 3. Renamings and substitutions on types

2.2 Renaming and Substitution

The most important operations on types are consistent renaming of type variables and substitution. Figure 3 contains the first step in defining renamings (left column) and substitutions (right column). Our definitions of renamings and substitutions in functional style are standard [1, 5, 9]. For now, we ignore the use of opaque. We defer the discussion of using it to Section 2.3.

A type renaming ζ maps variables from one scope n_1 to another n_2. We write the type of type renamings as Ren n_1 n_2. The definitions comprise (in order) weakening (wk^R), the identity renaming (id^R), extending a renaming with variable ($\alpha \cdot^R \zeta$), applying a renaming to a variable ($\alpha \,\&^R\, \zeta$), composing two renamings ($\zeta_1 \fatsemi^R \zeta_2$), lifting a renaming (to move under a binder) ($\zeta \uparrow^R$), and applying a renaming to a type ($T\,[\,\zeta\,]^R$).

For substitutions (right column of Figure 3) we proceed analogously. A type substitution η : Sub n_1 n_2 maps variables from scope n_1 to types defined over scope n_2. The definitions (in order) comprise lifting a renaming to a substitution ($\langle\ \zeta\ \rangle$), extending a substitution with a new type T ($T \cdot^S \eta$), applying a substitution to variable ($\alpha\ \&^S\ \eta$), lifting a substitution (to move under a binder) ($\eta \uparrow^S$), applying a substitution to a type ($T\ [\ \eta\ ]^S$), and composing two substitutions ($\eta_1 \fatsemi^S \eta_2$).

2.3 The σ-Calculus with First-class Renamings

This section presents the equalities that we *actually* used for computing with renamings and substitutions, rather than their defining equations in Figure 3. The opaque blocks enable us to control this behavior: Function symbols defined in an opaque block are usable outside the block, but they do not reduce! That is, the only reducing definitions in the renaming block are $\zeta \uparrow^R$ for lifting a renaming and application of a renaming to a type; in the substitution block, only the application of a substitution to a type reduces.

This selective reduction behavior is essential for our approach because we want to compute using the laws of the σ-calculus with first-class renamings on top of the thus defined symbols, rather than their definitions. Blocking their reduction is necessary because the equalities from σ-calculus interfere adversely with the standard definitional equalities. The original defining equalities are available for proofs: We can open the opaque definitions locally and proceed as usual.

Next we introduce equational laws of the σ-calculus with first-class renamings [11]. All of these equalities are supported by proofs using the definitions in Figure 3. The proofs are available in the supplement, but they are not included here for space reasons.

The first group presents computation rules for substitutions.

$$
\begin{array}{lll}
\text{beta-ext-zero} & : \mathsf{zero}\ \&^S\ (T \cdot^S \eta) & \equiv T \\
\text{beta-ext-suc} & : \mathsf{suc}\ \alpha\ \&^S\ (T \cdot^S \eta) & \equiv \alpha\ \&^S\ \eta \\
\text{beta-rename} & : \alpha\ \&^S\ \langle\ \zeta\ \rangle & \equiv\ \text{`}\ (\alpha\ \&^R\ \zeta) \\
\text{beta-comp} & : \alpha\ \&^S\ (\eta_1 \fatsemi^S \eta_2) & \equiv (\alpha\ \&^S\ \eta_1)\ [\ \eta_2\ ]^S \\
\text{beta-lift} & : \eta \uparrow^S & \equiv (\text{`}\ \mathsf{zero}) \cdot^S (\eta \fatsemi^S \langle\ \mathsf{wk}^R\ \rangle)
\end{array}
$$

Starting with the application of a substitution to a variable, there is one equation for each possible form of a substitution: extended substitution, lifted renaming, and composition of substitutions. The final equation in this group characterizes lifting in terms of extension, weakening, and composition. It is *not* possible to use this equation to define lifting in Figure 3 as it would lead to circularity.

The second group contains laws about interactions between substitutions.

$$
\begin{array}{lll}
\text{associativity} & : (\eta_1 \fatsemi^S \eta_2) \fatsemi^S \eta_3 & \equiv \eta_1 \fatsemi^S (\eta_2 \fatsemi^S \eta_3) \\
\text{distributivity} & : (T \cdot^S \eta_1) \fatsemi^S \eta_2 & \equiv (T\ [\ \eta_2\ ]^S) \cdot^S (\eta_1 \fatsemi^S \eta_2) \\
\text{interact} & : \langle\ \mathsf{wk}^R\ \rangle \fatsemi^S (T \cdot^S \eta) & \equiv \eta \\
\text{comp-id}_r & : \eta \fatsemi^S \langle\ \mathsf{id}^R\ \rangle & \equiv \eta \\
\text{comp-id}_l & : \langle\ \mathsf{id}^R\ \rangle \fatsemi^S \eta & \equiv \eta \\
\eta\text{-id} & : (\text{`}\ \mathsf{zero}\ \{n\}) \cdot^S \langle\ \mathsf{wk}^R\ \rangle & \equiv \langle\ \mathsf{id}^R\ \rangle \\
\eta\text{-law} & : (\mathsf{zero}\ \&^S\ \eta) \cdot^S (\langle\ \mathsf{wk}^R\ \rangle \fatsemi^S \eta) & \equiv \eta
\end{array}
$$

Composition is assoziative, it distributes over extension, weakening cancels extension, the identity substitution is a right and left identity, and we give two η-laws about the interaction of extension and weakening.[1]

[1] In the statement of η-id, we write zero $\{n\}$ to tell Agda's type checker that n is the scope of type variable zero. This notation supplies an *implicit argument* that Agda can usually infer.

```
lifts*-id : (id^S {n} ↑^S) ≡ id^S
lifts*-id = refl

lifts*-comp : (η' ⨟^S η) ↑^S ≡ (η' ↑^S) ⨟^S (η ↑^S)
lifts*-comp = refl
```

```
sub*-id : T [ id^S ]^S ≡ T
sub*-id = refl

sub*-var : (` α) [ η ]^S ≡ α &^S η
sub*-var = refl  - *

sub*-comp : T [ η ⨟^S η' ]^S ≡ (T [ η ]^S) [ η' ]^S
sub*-comp = refl - *
```

Fig. 4. Functor laws for substitution proved

There are analogous rules for renamings corresponding to the first two groups. These rules are omitted for space reasons.

The third group presents laws that govern the composition of different kinds of term traversal, that is, the behavior of the identity renaming and the four possible compositions of renamings and substitutions.

```
identity_r          : T [ id^R ]^R          ≡ T
compositionality^RS : (T [ ζ1 ]^R) [ η2 ]^S ≡ T [ ⟨ ζ1 ⟩ ⨟^S η2 ]^S
compositionality^RR : (T [ ζ1 ]^R) [ ζ2 ]^R ≡ T [ ζ1 ⨟^R ζ2 ]^R
compositionality^SR : (T [ η1 ]^S) [ ζ2 ]^R ≡ T [ η1 ⨟^S ⟨ ζ2 ⟩ ]^S
compositionality^SS : (T [ η1 ]^S) [ η2 ]^S ≡ T [ η1 ⨟^S η2 ]^S
```

The final group contains contains laws that mediate between substitutions and renamings: A substitution which applies a lifted renaming is just applying the renaming; the composition of two lifted renamings is the lifting of their composition as renamings.

```
coincidence      : T [ ⟨ ζ ⟩ ]^S      ≡ T [ ζ ]^R
coincidence-comp : ⟨ ζ1 ⟩ ⨟^S ⟨ ζ2 ⟩ ≡ ⟨ ζ1 ⨟^R ζ2 ⟩
```

In general ,the extraction of renamings from substitutions requires a dedicated solving strategy [13] and we have omitted some further coincidence laws for brevity.

Orienting these equations from left to right results in a confluent rewrite system for the σ-calculus that we install in Agda's computation engine using the REWRITE pragma [4]. Confluence of these rewrite equations is mechanically checked by Agda using the flags `-rewriting -confluence-check`. From now on, all equations shown in this section Section 2.3 are treated as definitional equalities.

As a consistency check for the definitions, we prove in Figure 4 that lifting a substitution and the action of a substitution satisfy the functor laws. With σ-calculus rewriting installed, all these lemmas hold definitionally as evidenced by their proof using reflexivity (refl). The lemmas marked with $*$ correspond directly to laws from the σ-calculus.[2] The lifting laws are shown on the left, the application laws on the right. The analogous results for renamings follow exactly the same pattern.

2.4 A Note on Safety

The theory underlying a proof assistant must be sound and consistent. Hence, it is a delicate act turning propositional equalities into definitional ones, for example by adding rewrite rules, because it can endanger both soundness and consistency. Recent work studies rewrite rules in

[2]The naming of the lemmas is taken from Chapman et al. [2] to enable direct comparison.

dependent type theories, including Agda [3] and Rocq [6]. Local rewrite rules are an active research direction [7].

Rewrite rules and the reductions of a dependently typed system interact in non-trivial ways. Hence, it is important to have criteria when it is safe to add rewrites. We justify making the equations from Section 2.3 definitional by appealing to the approach put forward for Rewriting Type Theory (RTT) [4], which is implemented in Agda. In RTT, the reduction relation is extended by turning propositional equalities into reduction rules: a term matched by the left-hand side of an equation reduces to the right-hand side, exactly like in Section 2.3. The two properties we wish to preserve when adding rewrite rules are subject reduction and consistency. In particular, the type checker assumes subject reduction and it would break without it.

To uphold subject reduction, the critical property of the newly added rewriting equations is that their *union with the existing reduction relation* is confluent. Agda already provides a standard set of confluent reduction rules: β, η (function and record expansion), δ (definition unfolding), and ι (dependent pattern matching clauses). Agda checks confluence of the extended system using a conservative set of simple syntactic criteria [4].

Some β and ι reductions arising from the functional definitions of renamings and substitutions in Figure 3 interfere with the intended rewrite equalities so that confluence breaks. For this reason, we explicitly hide these definitions inside an opaque block, so that only the equations we install with REWRITE define their reduction behaviour.

As a concrete example, consider the equation η-law: $(\mathsf{zero}\ \&^S\ \eta) \cdot^S (\langle\ \mathsf{wk}^R\ \rangle \fatsemi^S\ \eta) \equiv \eta$. Without the opaque block, the underlying definition $(\eta_1 \fatsemi^S \eta_2)\,\alpha = (\eta_1\ \alpha)\ [\ \eta_2\ ]^S$ would δ-unfold the inner composition $\langle\ \mathsf{wk}^R\ \rangle \fatsemi^S\ \eta$ into the lambda $\lambda\ \alpha \to \eta\ (\mathsf{suc}\ \alpha)$, so that the entire left-hand side reduces to $(\eta\ \mathsf{zero}) \cdot^S (\lambda\ \alpha \to \eta\ (\mathsf{suc}\ \alpha))$. Two issues arise. First, the rewrite engine would no longer be able to match the left-hand side of η-law syntactically during reduction[3]. Second, even if the rewrite rule were admitted, confluence would still fail. The two reducts η (via the rewrite) and $(\eta\ \mathsf{zero}) \cdot^S (\lambda\ \alpha \to \eta\ (\mathsf{suc}\ \alpha))$ (via δ-unfolding) are propositionally equal but not definitionally equal. Because the function $\cdot^S$ case-splits on a neutral variable, Agda cannot reduce the second reduct back to η. Hiding the definition of the symbol $\fatsemi^S$ (together with $\cdot^S$ and the other operators) inside an opaque block turns them into stuck symbols, restoring both matchability and confluence.

Consistency is preserved as long as the equalities we rewrite are instantiated with proofs. This is easy to see, because RTT can be translated to extensional type theory [4], a theory known to be consistent.

Agda currently supports all features we require: opaque-defined symbols, proof-instantiated rewrite equations, and triangle-based confluence checking. Rocq currently lacks the first two, although the overall method otherwise transfers.

3 System F Expressions

3.1 Syntax

Working towards intrinsically-typed expressions, Figure 5 defines type weakening (weaken) and single substitution $\mathsf{T}\ [\ \mathsf{T}'\ ]^*$ of a type T' for the innermost variable of T along with type contexts Γ, expression variables x, and the syntax of expressions e. Type contexts and expression variables follow the work of Chapman et al. [2]. A type context $\Gamma : \mathsf{Ctx}\ n$ tracks expression and type variables in scope n. The operator $\_\triangleright\_$ adds an value binding while $\_\triangleright^*$ adds a type binding to Γ. Correspondingly, expression variables are implemented as typed de Bruijn indices with an extra

[3] To see why, note that η does not appear in pattern position here. A precise definition of pattern position is available in the Agda Wiki: https://agda.readthedocs.io/en/latest/language/rewriting.html#general-shape-of-rewrite-rules

```
weaken : Type n → Type (1 + n)
weaken T = T [ wk^R ]^R

_[_]* : Type (1 + n) → Type n → Type n
T [ T′ ]* = T [ T′ ·^S id^S ]^S

data Ctx : Nat → Set where
  ∅   : Ctx zero
  _▷_ : Ctx n → Type n → Ctx n
  _▷* : Ctx n → Ctx (1 + n)

data _∋_ : Ctx n → Type n → Set where
  zero : (Γ ▷ T) ∋ T
  suc  : Γ ∋ T → (Γ ▷ T′) ∋ T
  suc* : Γ ∋ T → (Γ ▷*) ∋ weaken T
```

```
data Expr (Γ : Ctx n) : Type n → Set where
  `_  : Γ ∋ T →
        Expr Γ T
  λx  : Expr (Γ ▷ T₁) T₂ →
        Expr Γ (T₁ ⇒ T₂)
  _·_ : Expr Γ (T₁ ⇒ T₂) →
        Expr Γ T₁ →
        Expr Γ T₂
  Λα  : Expr (Γ ▷*) T →
        Expr Γ (∀α T)
  _·*_ : Expr Γ (∀α T) →
         (T′ : Type n) →
         Expr Γ (T [ T′ ]*)
```

Fig. 5. Type contexts, variables, and expressions

case suc* to skip over type variables in the context. Skipping requires weakening of the type as it is used under one more type binding.

Expressions e : Expr Γ T are indexed by a context Γ and a type T with the intended reading as a typing judgment $\Gamma \vdash e : T$. As customary in an intrinsically typed setting, each expression constructor corresponds to a typing rule in System F. The rule for type application relies on single substitution for types.

3.2 Examples

As an example of System F expressions, we present some definitions for Church numerals. We start with their type and the encoding of zero. For readability, we define aliases for type variables like `α for ` Z, `β for ` (S Z), and analogously for expression variables and binders.

```
ℕ^c : Type ∅ *
ℕ^c = ∀α ((`α ⇒ `α) ⇒ (`α ⇒ `α))

zero^c : Expr ∅ ℕ^c
zero^c = Λα (λg (λx `x))
```

We continue with the constant one and the Church numeral for two.

```
one^c : Expr ∅ ℕ^c
one^c = Λα (λg (λx (`g · `x)))

two^c : Expr ∅ ℕ^c
two^c = succ^c · (succ^c · zero^c)
```

The successor operation has a longer body.

```
succ^c : Expr ∅ (ℕ^c ⇒ ℕ^c)
succ^c = λx (Λα (λg (λx (`g · (((( ` (suc (suc (suc* zero)))) ·* `α) · `g) · `x)))))
```

3.3 Renaming and Substitution

The definition of a small-step operational semantics requires expression renamings and substitution to implement beta reduction for value abstractions and type abstractions. Renamings and substitutions are both indexed by type renamings and type substitutions, respectively. We omit

renamings here, as their development is analogous to the one for substitutions and concentrate on explaining the latter.

Expression substitutions σ are indexed by type substitutions η. A substitution takes a variable x of type T in context Γ_1 and maps it to an expression typed in context Γ_2 adjusting its type according to the type substitution: $T[\eta]^S$. Notationally, we write $\eta \mid \ldots$ for any operation on substitutions that is indexed by type substitution η.

```
_|_⇒ˢ_ : Sub n₁ n₂ → Ctx n₁ → Ctx n₂ → Set
η | Γ₁ ⇒ˢ Γ₂ = ∀ T → (x : Γ₁ ∋ T) → Expr Γ₂ (T [ η ]ˢ)
```

Applying a substitution to a variable corresponds to function application. Nevertheless, we use the same interface as for type substitutions to enable using the same rewriting machinery on the expression level again. For this definition, we include the typing of all symbols, which is similar for all further definitions and omitted because it can be inferred.

```
_|_&ˢ_ : ∀ {Γ₁ : Ctx n₁} {Γ₂ : Ctx n₂} (η : Sub n₁ n₂)
  → (x : Γ₁ ∋ T) → (σ : η | Γ₁ ⇒ˢ Γ₂) → Expr Γ₂ (T [ η ]ˢ)
η | x &ˢ σ = σ _ x
```

The identity substitution for expressions is straightforward to define. But this simplicity is due to the installed rewriting of type substitutions. Without the rewriting, the right hand side's type is T, but the expected type is $T[\mathsf{id}^s]^S$, which would require the transport lemma $\mathsf{comp\text{-}id}_r$.

```
Idˢ : idˢ | Γ ⇒ˢ Γ
Idˢ _ = λ x → ‘ x
```

Extending a substitution is standard: given an expression e and a substitution σ, we return a substitution that sends variable zero to e and otherwise resorts to σ.

```
_|_·ˢ_ : ∀ η → (e : Expr Γ₂ (T [ η ]ˢ)) → (σ : η | Γ₁ ⇒ˢ Γ₂) → η | (Γ₁ ▷ T) ⇒ˢ Γ₂
(_ | e ·ˢ σ) _ zero    = e
(_ | e ·ˢ σ) _ (suc x) = σ _ x
```

Due to the structure of contexts, we need two lifting lemmas when defining the application of a substitution to an expression, one for value bindings and another for type bindings. In both cases, the substitution σ needs to be adjusted (i.e., lifted) to accommodate the newly introduced binding. The former is standard: it maps the newly introduced variable to itself (zero) and weakens the images of σ accordingly. However, the definition of lifting requires a transport lemma to line up the type substitutions, which is applied by rewriting. The last part of the definition $\mathsf{id}^R \mid \ldots \mathsf{Wk}^R \ldots$ applies an expression renaming (weakening by $T[\eta]^S$) indexed by id^R to the output of σ.

```
_|_↑ˢ_ : ∀ η → (σ : η | Γ₁ ⇒ˢ Γ₂) → ∀ T → η | (Γ₁ ▷ T) ⇒ˢ (Γ₂ ▷ (T [ η ]ˢ))
η | σ ↑ˢ T = η | (‘ zero) ·ˢ λ _ x → idᴿ | (σ _ x) [ Wkᴿ (T [ η ]ˢ) ]ᴿ
```

Lifting a substitution over a type binding is non-standard, but straightforward. The type substitution requires lifting, but no rewriting is needed.

```
_|_↑ˢ* : ∀ η → (σ : η | Γ₁ ⇒ˢ Γ₂) → (η ↑ˢ) | (Γ₁ ▷*) ⇒ˢ (Γ₂ ▷*)
(η | σ ↑ˢ*) = (η ⨟ˢ ⟨ wkᴿ ⟩) | (‘ zero) ·ˢ* λ _ x → wkᴿ | (σ _ x) [ wkᴿ* ]ᴿ
```

Substitution application is defined by structural recursion on expressions. The cases dispatch to the operations that we introduced: applying a substitution to a variable, lifting over a value binding (case λx), lifting over a type binding (case $\Lambda\alpha$), the cases for application and type-application are homomorphic.

```
_|_[_]ˢ : (η : Sub n₁ n₂) → (e : Expr Γ₁ T) → (σ : η | Γ₁ ⇒ˢ Γ₂) → Expr Γ₂ (T [ η ]ˢ)
η | (` x) [ σ ]ˢ     = η | x &ˢ σ
η | (λx e) [ σ ]ˢ    = λx (η | e [ η | σ ⇑ˢ _ ]ˢ)
η | (Λα e) [ σ ]ˢ    = Λα ((η ↑ˢ) | e [ η | σ ⇑ˢ* ]ˢ)
η | (e · e₁) [ σ ]ˢ  = (η | e [ σ ]ˢ) · (η | e₁ [ σ ]ˢ)
η | (e ·* T) [ σ ]ˢ = (η | e [ σ ]ˢ) ·* (T [ η ]ˢ)
```

3.4 Semantics

We are now ready to define a small-step operational semantics. As in the introduction, we opt for call-by-name for concision. The definition of single substitution, needed for beta reduction, is analogous to the one on types: we build the substitution by extending the identity substitution with the argument e'. As usual in an intrinsically typed setting, this definition already proves that substitution preserves typing!

```
_[_] : Expr (Γ ▷ T′) T → Expr Γ T′ → Expr Γ T
e [ e′ ] = idˢ | e [ idˢ | e′ ·ˢ Idˢ ]ˢ
```

In addition, we have to substitute a type into an expression to cater for beta reduction of type applications. This substitution is also implemented by an expression substitution indexed by a type substitution that performs the actual change.

```
_[*_*] : Expr (Γ ▷*) T → (T′ : Type n) → Expr Γ (T [ T′ ]*)
e [* T′ *] = (T′ ·ˢ idˢ) | e [ idˢ | T′ ·ˢ* Idˢ ]ˢ
```

Due to the extended type substitution on the left, we also have to adapt the expression substitution on the right to the extended context structure. This adaptation is the type-level analogue to extension:

```
_|_·ˢ*_ : ∀ η T → (σ : η | Γ₁ ⇒ˢ Γ₂) → (T ·ˢ η) | (Γ₁ ▷*) ⇒ˢ Γ₂
(_ | T ·ˢ* σ) _ (suc* x) = σ _ x
```

With substitution in place, the definition of the call-by-name small-step reduction relation is straightforward. As usual, such a definition of reduction for intrinsically-typed syntax implies subject reduction by construction.

```
data _⟶_ : Expr Γ T → Expr Γ T → Set where
  β-λ : (λx e₁ · e₂)  ⟶ (e₁ [ e₂ ])
  β-Λ : (Λα e ·* T′) ⟶ (e [* T′ *])
  ξ-·  : e₁ ⟶ e₁′ → (e₁ · e₂) ⟶ (e₁′ · e₂)
  ξ-·* : e ⟶ e′   → (e ·* T) ⟶ (e′ ·* T)
  ξ-Λ : e ⟶ e′   → (Λα e) ⟶ (Λα e′)
```

To obtain type safety, it remains to prove progress. Figure 6 presents the complete proof of progress for call-by-name System F using the structure from PLFA [5]. A value is either a lambda or a big lambda where the body is already a value. An expression satisfies progress if it is either a value (done) or if it can make a step (step). Due to the definition of value for big lambdas, we have to establish progress for contexts that may contain type bindings, but no value bindings. These contexts are characterized literally by the function NoVar. The actual proof of progress is by induction on the structure of expressions. The overall structure of this proof closely follows Chapman et al [2]. Their proof is more complicated because their semantics is call-by-value, their language supports isorecursive types, and their subject language is System Fω.

```
data Value : Expr Γ T → Set where
  λx  : (e : Expr (Γ ▷ T₁) T₂) → Value (λx e)
  Λα : (v : Value e)           → Value (Λα e)

data Progress : Expr Γ T → Set where
  done : (v : Value e) → Progress e
  step : (e⟶e′ : e ⟶ e′) → Progress e

NoVar : Ctx n → Set
NoVar Γ = ∀ {T′} → ¬ (Γ ∋ T′)

progress : NoVar Γ → (e : Expr Γ T) → Progress e
progress nv (` x)   = ⊥-elim (nv x)
progress nv (λx e) = done (λx e)
progress nv (e · e₁)
  with progress nv e
... | done (λx e₂)    = step β-λ
... | step e⟶e′ = step (ξ-· e⟶e′)
progress nv (Λα e)
  with progress (λ{ (suc* x) → nv x}) e
... | done v        = done (Λα v)
... | step e⟶e′ = step (ξ-Λ e⟶e′)
progress nv (e ·* T′)
  with progress nv e
... | done (Λα v) = step β-Λ
... | step e⟶e′ = step (ξ-·* e⟶e′)
```

Fig. 6. Progress for System F

4 Transfer Heaven and Transfer Hell

Proving type safety for System F requires exercising the substitution theory at the type level. Without installing its laws as definitional equalities, many definitions become cumbersome because transport lemmas have to be inserted to equalize the types of (Agda) expressions.

As we have seen in Section 3.4, proving type safety does *not* require exercising substitution theory at the expression level. However, as a sanity check of the definitions we verify functorial and monadic properties of expression substitutions. Moreover, if we were to move on to constructing and proving properties of logical relations, then these properties would be needed [10].

4.1 Transfer Heaven: Composition of Substitutions with Rewriting

As an example, we examine the composition of two expression substitutions. Its definition involves two type substitutions η_1 and η_2 that index the corresponding expression substitutions σ_1 and σ_2. The composed expression substitution is indexed by the composition of η_1 and η_2. Type checking this definition already exercises the laws for composition of type substitutions.

```
_,_|_⨟ˢ_ : ∀ η₁ η₂ → (σ₁ : η₁ | Γ₁ ⇒ˢ Γ₂) → (σ₂ : η₂ | Γ₂ ⇒ˢ Γ₃) → (η₁ ⨟ˢ η₂) | Γ₁ ⇒ˢ Γ₃
(η₁ , η₂ | σ₁ ⨟ˢ σ₂) _ x = η₂ | (σ₁ _ x) [ σ₂ ]ˢ
```

We state the functoriality property of expression substitution: applying two substitutions one after the other to an expression is the same as applying their composition. This statement is a very prominent example of transfer hell: the types of the left and right side of the equality are different and would require a transfer without our rewriting machinery in place.

```
Compositionalityˢˢ : ∀ (e : Expr Γ₁ T) (η₁ : Sub n₁ n₂) (η₂ : Sub n₂ n₃)
  → (σ₁ : η₁ | Γ₁ ⇒ˢ Γ₂) (σ₂ : η₂ | Γ₂ ⇒ˢ Γ₃)
  → η₂ | (η₁ | e [ σ₁ ]ˢ) [ σ₂ ]ˢ ≡ (η₁ ⨟ˢ η₂) | e [ η₁ , η₂ | σ₁ ⨟ˢ σ₂ ]ˢ
```

We show the proof of this property not as a subject of study, but to demonstrate that transfers for type substitutions are not needed. We can concentrate on the gist of the proof, rather than being overwhelmed by transfers. The remaining complication is the explicit passing of the type

substitutions η_1 and η_2. We chose to make them explicit; while Agda can infer them in many places, there are equally many places where inference is not possible.

```
Compositionality^SS (` x)       η₁ η₂ σ₁ σ₂ = refl
Compositionality^SS (λx {T₁ = T₁} e) η₁ η₂ σ₁ σ₂ = cong λx (begin
    η₂ ∣ η₁ ∣ e [ η₁ ∣ σ₁ ⇑^S T₁ ]^S [ η₂ ∣ σ₂ ⇑^S (T₁ [ η₁ ]^S) ]^S
  ≡⟨ Compositionality^SS e η₁ η₂ (η₁ ∣ σ₁ ⇑^S T₁) (η₂ ∣ σ₂ ⇑^S (T₁ [ η₁ ]^S)) ⟩
    (η₁ ⨟^S η₂) ∣ e [ η₁ , η₂ ∣ η₁ ∣ σ₁ ⇑^S T₁ ⨟^S (η₂ ∣ σ₂ ⇑^S (T₁ [ η₁ ]^S)) ]^S
  ≡⟨ cong ((η₁ ⨟^S η₂) ∣ e [_]^S) (Lift-Dist-Comp^SS σ₁ σ₂) ⟩
    (η₁ ⨟^S η₂) ∣ e [ (η₁ ⨟^S η₂) ∣ η₁ , η₂ ∣ σ₁ ⨟^S σ₂ ⇑^S T₁ ]^S
  □)
Compositionality^SS (Λα e)   η₁ η₂ σ₁ σ₂ = cong Λα (begin
    (η₂ ↑^S) ∣ (η₁ ↑^S) ∣ e [ η₁ ∣ σ₁ ⇑^S* ]^S [ η₂ ∣ σ₂ ⇑^S* ]^S
  ≡⟨ Compositionality^SS e (η₁ ↑^S) (η₂ ↑^S) (η₁ ∣ σ₁ ⇑^S*) (η₂ ∣ σ₂ ⇑^S*) ⟩
    ((η₁ ⨟^S η₂) ↑^S) ∣ e [ (η₁ ↑^S) , η₂ ↑^S ∣ η₁ ∣ σ₁ ⇑^S* ⨟^S (η₂ ∣ σ₂ ⇑^S*) ]^S
  ≡⟨ cong (((η₁ ⨟^S η₂) ↑^S) ∣ e [_]^S) (Lift*-Dist-Comp^SS η₁ η₂ σ₁ σ₂) ⟩
    ((η₁ ⨟^S η₂) ↑^S) ∣ e [ (η₁ ⨟^S η₂) ∣ η₁ , η₂ ∣ σ₁ ⨟^S σ₂ ⇑^S* ]^S
  □)
Compositionality^SS (e₁ · e₂) η₁ η₂ σ₁ σ₂ = cong₂ _·_
  (Compositionality^SS e₁ η₁ η₂ σ₁ σ₂)
  (Compositionality^SS e₂ η₁ η₂ σ₁ σ₂)
Compositionality^SS (e ·* T′) η₁ η₂ σ₁ σ₂ = cong (_·* (T′ [ η₁ ⨟^S η₂ ]^S))
  (Compositionality^SS e η₁ η₂ σ₁ σ₂)
```

This proof relies on further properties, in particular Lift-Dist-Comp^{SS} (lifting a composition of substitutions is equal to composing their liftings) and Lift*-Dist-Comp^{SS} (its counterpart for lifting over type bindings), where the statement and proof would require explicit transfers. Essentially, these properties form a "food chain" that stops at the composition properties of renamings with renamings and liftings, all of which work similarly. The supplemental material contains full details.

4.2 Transfer Hell: Composition of Substitutions Without Rewriting

We now contrast the development from Section 4.1 with the same proof carried out *without* installing the σ-calculus equations as definitional equalities. As a point of comparison, we take the supplement of the construction of a logical relation [10], which also mechanizes the compositionality properties of expression substitutions for System F. To enable a direct comparison, we have aligned the surface syntax of their definitions with ours.

The type of compositionality already exposes the core issue. Because the σ-calculus equations are only propositional, the theorem statement must include subst to bridge indices that are equal but not definitionally equal : the type of the left side of the equality is *not* definitionally equal to the type of the right side. To adapt the types, compositionality of type substitutions is applied to the type of the expression using function sub.

```
Compositionality^SS-≡ :
  ∀ {η₁ : n₁ →^S n₂} {η₂ : n₂ →^S n₃}
    {Γ₁ : Ctx n₁} {Γ₂ : Ctx n₂} {Γ₃ : Ctx n₃}
    (e : Expr n₁ Γ₁ T)
    (σ₁ : η₁ ∣ Γ₁ ⇒^S Γ₂) (σ₂ : η₂ ∣ Γ₂ ⇒^S Γ₃) →
```

```
  let sub = subst (Expr n₃ Γ₃) (compositionality^SS T η₁ η₂) in
  sub (η₂ ⁏ (η₁ ⁏ e [ σ₁ ]^S) [ σ₂ ]^S) ≡ (η₁ ⨟^S η₂) ⁏ e [ η₁ , η₂ ⁏ σ₁ ⨟^S σ₂ ]^S
```

Because subst is part of the statement, every recursive call to the induction hypothesis introduces a fresh application of subst, which has to be discharged in the proof alongside the interesting proof steps. To do so, it must be distributed to the outside of the term and combined with the substs arising from neighboring subterms before the goal can be closed. This bookkeeping is further complicated by the fact that the other appearances of type substitution lemmas required in the proof interact with the transport equalities in non-trivial ways. We illustrate the resulting complexity by zooming in on a single case of the inductive proof—type application—of their compositionality lemma.

Firstly, our development already saves one subst in the very definition of substitution application. The type application case requires the swap lemma of type $(T\,[\,T'\,]^*)\,[\,\eta\,] \equiv (T\,[\,\eta \uparrow^S\,])\,[\,T'\,[\,\eta\,]\,]^*$ to coerce the term to the expected type, leading to the following definition.

```
η ⁏ (_ ·* _ {T = T} e T') [ σ ]^S = subst (Expr _ _)
                                    (sym (swap^SS η T T'))
                                    ((η ⁏ e [ σ ]^S) ·* (T' [ η ]^S))
```

Next, we examine the type application case of the compositionality proof itself. The authors employ heterogeneous equality [8] to avoid some of the administrative reasoning otherwise forced by transfer hell. Without heterogeneous equality, the proof would be even more complex, requiring several auxiliary lemmas that explicitly shuffle applications of subst around. Thus, the version below already constitutes one of the simplest forms of the argument.

```
let
  F₂ = Expr n₂ Γ₂ ; E₂ = sym (swap^SS η₁ T T')          ; sub₂ = subst F₂ E₂
  F₃ = Expr n₃ Γ₃ ; E₃ = sym (swap^SS (η₁ ⨟^S η₂) T T') ; sub₃ = subst F₃ E₃
  F₅ = Expr n₃ Γ₃ ; E₅ = sym (swap^SS η₂ (T [ η₁ ↑^S ]) (T' [ η₁ ]^S)) ; sub₅ = subst F₅ E₅
in
R.begin
  η₂ ⁏ (η₁ ⁏ (e ·* T') [ σ₁ ]^S) [ σ₂ ]^S
R.≅⟨ refl ⟩                                                          - (1)
  η₂ ⁏ (sub₂ ((η₁ ⁏ e [ σ₁ ]^S) ·* (T' [ η₁ ]^S))) [ σ₂ ]^S
R.≅⟨ H.cong₂ {B = Expr n₂ Γ₂} (λ _ ■ → η₂ ⁏ ■ [ σ₂ ]^S)              - (2)
       (H.≡-to-≅ (sym E₂))
       (H.≡-subst-removable F₂ E₂ _) ⟩
  η₂ ⁏ ((η₁ ⁏ e [ σ₁ ]^S) ·* (T' [ η₁ ]^S)) [ σ₂ ]^S
R.≅⟨ refl ⟩                                                          - (3)
  sub₅ ((η₂ ⁏ (η₁ ⁏ e [ σ₁ ]^S) [ σ₂ ]^S) ·* (T' [ η₁ ]^S [ η₂ ]^S))
R.≅⟨ H.≡-subst-removable F₅ E₅ _ ⟩                                   - (4)
  (η₂ ⁏ (η₁ ⁏ e [ σ₁ ]^S) [ σ₂ ]^S) ·* (T' [ η₁ ]^S [ η₂ ]^S)
R.≅⟨ Hcong₃ {B = λ ■ → Expr n₃ Γ₃ (∀α ■)} {C = λ _ _ → Type n₃ _ }  - (5)
       (λ _ ■₁ ■₂ → ■₁ ·* ■₂)
       (H.≡-to-≅ (lift-compositionality^SS T η₁ η₂))
       (Compositionality^SS e σ₁ σ₂)
       (H.≡-to-≅ (compositionality^SS T' η₁ η₂)) ⟩
  ((η₁ ⨟^S η₂) ⁏ e [ η₁ , η₂ ⁏ σ₁ ⨟^S σ₂ ]^S) ·* (T' [ η₁ ⨟^S η₂ ]^S)
```

	With Rewriting		Transfer Hell	
	LOC	Chars	LOC	Chars
Type substitution + proofs	217	12 246	285	12 584
Expression substitution + proofs	199	11 465	1 405	69 846
Total	**416**	**23 711**	**1 690**	**82 430**
Type-check time	4.49 s		9.51 s	

Table 1. Proof statistics

```
R.≅⟨ H.sym (H.≡-subst-removable F₃ E₃ _) ⟩                    - (6)
  sub₃ (((η₁ ⨟ˢ η₂) ∣ e [ η₁ , η₂ ∣ σ₁ ⨟ˢ σ₂ ]ˢ) ·* (T′ [ η₁ ⨟ˢ η₂ ]ˢ))
R.≅⟨ refl ⟩                                                   - (7)
  (η₁ ⨟ˢ η₂) ∣ (e ·* T′) [ η₁ , η₂ ∣ σ₁ ⨟ˢ σ₂ ]ˢ
R.□
```

The proof proceeds in seven steps.

(1) Unfold the definition of expression substitution at the inner application of σ_1, exposing the subst inside.
(2) Remove the subst that sits inside the term under consideration, justified by heterogeneous equality.
(3) Reveal yet another subst introduced by the substitution application of σ_2
(4) Remove the subst on the outside.
(5) Apply the induction hypothesis while simultaneously coercing the index according to the compositionality law for type substitutions.
(6) Reinsert the subst demanded by the proof goal on the outside.
(7) Fold the definition of expression substitution to conclude the proof.

Of these, steps (2), (4), the type coercion in (5), and (6), as well as the use of heterogeneous equality, are entirely absent from our rewriting-based proof, only the application of the inductive hypothesis and the implicit (un)folding of definitions remain.

We complement this qualitative comparison with a quantitative one. Table 1 records the lines of code and number of characters required to mechanize the compositionality law for expression substitutions, together with the type-checking time of the respective developments.

As expected, the amount of code required to formalize type substitution is roughly the same in both developments. For expression substitutions, however, the picture changes: the transfer-hell version requires more than five times as many lines of code as the rewriting-based one. As a welcome side effect, the type-checking time is also reduced by about 50% when the σ-calculus equations are installed as rewrite rules.

5 Going Further

5.1 Equational Theory for Expression Substitution

Having proven the functor laws for expression substitution, one of the hardest results has already been established. A natural question is whether we can go further and install a full equational theory for expression substitution as definitional equalities, just as we did at the type level. We show that it is possible. The σ-calculus for type-substitution-indexed expression substitutions is an extension of the one for type substitutions in Section 2.3, and the same holds analogously for expression renamings.

Expression substitutions carry additional operators that have no counterpart at the type level. Every law involving extension or weakening splits into two expression-level laws. One goversn the expression variants ($\_\cdot^S\_$, wk^S) and directly mirrors the type-level rule Section 2.3. The other goversn the type variants ($\_\cdot^{S*}\_$, wk^{S*}) and is genuinely new. We illustrate this pattern with the η-id law. The expression variant

$$\eta\text{-}\mathsf{Id}^S : \langle\ \mathsf{id}^R\ \rangle \mid (\text{`}\ \mathsf{zero}) \cdot^S (\mathsf{Wk}^S\ T) \equiv \mathsf{Id}^S\ \{\Gamma = \Gamma \triangleright\ T\}$$

is a direct transcription of η-id from Section 2.3, whereas the type variant

$$\eta\text{-}\mathsf{Id}^{S*} : \langle\ \mathsf{wk}^R\ \rangle \mid (\text{`}\ \mathsf{zero}) \cdot^{S*} \mathsf{wk}^{S*} \equiv \mathsf{Id}^S\ \{\Gamma = \Gamma \triangleright^*\}$$

asserts that extending the type-binding weakening with the topmost type variable ‘ zero yields the identity expression substitution. The same doubling applies to the remaining laws in which extension or weakening appears (η, interact, distributivity and some β laws).

The full equational system is included in the supplement, and Agda verifies its confluence. For the confluence check to go through, we had to artificially block reduction of the traversal functions $\_|\_[\_]^R$ and $\_|\_[\_]^S$ and re-install each of their defining clauses as a rewrite rule. In the type application case, rewriting must perform two reduction steps at once to satisfy the triangle criterion. One reduction normalizes the type index; the other pushes substitution into the expression.

5.2 Practical Guidelines for Other Intrinsic Developments

We have now made the equations of the σ-calculus definitional at two levels of the syntactic hierarchy: at the type level and at the expression level, where the latter depends on the former. In general, this strategy applies to arbitrarily tall towers of intrinsically typed syntaxes, in which each level's type indices themselves support substitution.

More precisely, consider a tower $T_0, T_1, \ldots, T_n$ of syntactic sorts in which the variables and substitutions of T_{i+1} are indexed by the contexts and substitutions of $T_0, \ldots, T_i$. The recipe proceeds bottom-up:

(1) Define T_0 together with its renamings and substitutions, and install the σ-calculus equations for T_0 as rewrite rules.
(2) For each i from 1 to n, define T_i *on top of* the already-definitional substitution machinery of $T_0, \ldots, T_{i-1}$, prove the corresponding σ-calculus laws for T_i, and install them as rewrite rules.

The key is that the laws for all levels $< i$ are already definitional, when we define and prove the laws for level i. Otherwise, the types of the level-i operations would no longer match definitionally, and we would be back in transfer hell.

The size of the equational theory grows with each level. A substitution at level i can be lifted under a binder of any sort $T_0, \ldots, T_{i-1}$, and each kind of binder induces its own family of laws, for weakening and extension, and the corresponding interactions with composition and traversal. At the type level this contributes no additional laws. At the expression level we already saw a few new laws, namely the laws governing weakening and extension by types. At level three one would add even more laws, and so on. The rule count therefore grows linearly with the level, but each new family follows the same template as the previous one, so the proof effort per family stays constant.

As a concrete instance, the language System $\mathrm{F}^{\uparrow}_{C}$ [14] features three levels: kinds, types, and expressions. Each level carries its own variables and substitution. Using our approach, it should be possible to construct an intrinsically typed representation for it and establish the substitution theory for all three levels uniformly, without any transfer reasoning.

6 Conclusion

Intrinsically typed syntax facilitates elegant mechanizations by making well-typed programs representable only as well-typed data. For polymorphic calculi this promise is often undermined by pervasive transport reasoning. Our experience with System F shows that the difficulty does not stem from binding structure or indexing itself, but from a mismatch between substitution and definitional equality: substitutions that are extensionally equal fail to compute to a canonical form. By internalizing the equational laws of the σ-calculus as definitional equalities, substitutions become canonical and routine transports disappear from proofs. The resulting developments recover the elegance seen in the simply-typed case while scaling to towers of syntactic sorts. More broadly, this fresh perspective suggests a practical guideline for mechanizing type systems in dependently typed proof assistants: when intrinsically typed encodings become proof-heavy, a promising strategy is often to strengthen definitional equality rather than introduce additional lemmas or automation.

References

[1] Nick Benton, Chung-Kil Hur, Andrew Kennedy, and Conor McBride. 2012. Strongly Typed Term Representations in Coq. *J. Autom. Reason.* 49, 2 (2012), 141–159. doi:10.1007/S10817-011-9219-0

[2] James Chapman, Roman Kireev, Chad Nester, and Philip Wadler. 2019. System F in Agda, for Fun and Profit. In *Mathematics of Program Construction: 13th International Conference, MPC 2019, Porto, Portugal, October 7–9, 2019, Proceedings* (Porto, Portugal). Springer-Verlag, Berlin, Heidelberg, 255–297. doi:10.1007/978-3-030-33636-3_10

[3] Jesper Cockx. 2020. Type Theory Unchained: Extending Agda with User-Defined Rewrite Rules. In *25th International Conference on Types for Proofs and Programs (TYPES 2019) (Leibniz International Proceedings in Informatics (LIPIcs), Vol. 175)*, Marc Bezem and Assia Mahboubi (Eds.). Schloss Dagstuhl – Leibniz-Zentrum für Informatik, Dagstuhl, Germany, 2:1–2:27. doi:10.4230/LIPIcs.TYPES.2019.2

[4] Jesper Cockx, Nicolas Tabareau, and Théo Winterhalter. 2021. The Taming of the Rew: A Type Theory with Computational Assumptions. *Proc. ACM Program. Lang.* 5, POPL, Article 60 (Jan. 2021), 29 pages. doi:10.1145/3434341

[5] Wen Kokke, Jeremy G. Siek, and Philip Wadler. 2020. Programming Language Foundations in Agda. *Sci. Comput. Program.* 194 (2020), 102440. doi:10.1016/J.SCICO.2020.102440

[6] Yann Leray, Gaëtan Gilbert, Nicolas Tabareau, and Théo Winterhalter. 2024. The Rewster: Type Preserving Rewrite Rules for the Coq Proof Assistant. In *15th International Conference on Interactive Theorem Proving (ITP 2024) (Leibniz International Proceedings in Informatics (LIPIcs), Vol. 309)*, Yves Bertot, Temur Kutsia, and Michael Norrish (Eds.). Schloss Dagstuhl – Leibniz-Zentrum für Informatik, Dagstuhl, Germany, 26:1–26:18. doi:10.4230/LIPIcs.ITP.2024.26

[7] Yann Leray and Théo Winterhalter. 2026. Encode the Cake and Eat It Too: Controlling Computation in Type Theory, Locally. *Proc. ACM Program. Lang.* 10, POPL, Article 62 (Jan. 2026), 30 pages. doi:10.1145/3776704

[8] Conor McBride. 2000. Elimination with a Motive. In *Types for Proofs and Programs, International Workshop, TYPES 2000, Durham, UK, December 8-12, 2000, Selected Papers (Lecture Notes in Computer Science, Vol. 2277)*, Paul Callaghan, Zhaohui Luo, James McKinna, and Robert Pollack (Eds.). Springer, 197–216. doi:10.1007/3-540-45842-5_13

[9] Conor McBride. 2005. Type-preserving Renaming and Substitution. (2005). http://strictlypositive.org/ren-sub.pdf Unpublished manuscript.

[10] Hannes Saffrich, Peter Thiemann, and Marius Weidner. 2024. Intrinsically Typed Syntax, a Logical Relation, and the Scourge of the Transfer Lemma. In *Proceedings of the 9th ACM SIGPLAN International Workshop on Type-Driven Development* (Milan, Italy) *(TyDe 2024)*. Association for Computing Machinery, New York, NY, USA, 2–15. doi:10.1145/3678000.3678201

[11] Steven Schäfer, Tobias Tebbi, and Gert Smolka. 2015. Autosubst: Reasoning with de Bruijn Terms and Parallel Substitutions. In *Interactive Theorem Proving*, Christian Urban and Xingyuan Zhang (Eds.). Springer International Publishing, Cham, 359–374. https://www.ps.uni-saarland.de/Publications/documents/SchaeferEtAl_2015_Autosubst_-Reasoning.pdf

[12] Kathrin Stark. 2020. *Mechanising Syntax with Binders in Coq.* Ph. D. Dissertation. Saarland University, Saarbrücken, Germany. https://www.ps.uni-saarland.de/Publications/documents/Stark_2020_Mechanising.pdf

[13] Kathrin Stark, Steven Schäfer, and Jonas Kaiser. 2019. Autosubst 2: Reasoning with Multi-sorted de Bruijn Terms and Vector Substitutions *(CPP 2019)*. Association for Computing Machinery, New York, NY, USA, 166–180. doi:10.1145/3293880.3294101

[14] Brent A. Yorgey, Stephanie Weirich, Julien Cretin, Simon Peyton Jones, Dimitrios Vytiniotis, and José Pedro Magalhães. 2012. Giving Haskell a Promotion. In *Proceedings of the 8th ACM SIGPLAN Workshop on Types in Language Design and Implementation* (Philadelphia, Pennsylvania, USA) *(TLDI '12)*. Association for Computing Machinery, New York, NY,

USA, 53–66. doi:10.1145/2103786.2103795